NASA Technical Memorandum 83794

Ultrasonic Velocity Measurement Using Phase-Slope and Cross-Correlation Methods

David R. Hull, Harold E. Kautz,
and Alex Vary
Lewis Research Center
Cleveland, Ohio

Prepared for the
1984 Spring Conference of the American
Society for Nondestructive Testing
Denver, Colorado, May 21-24, 1984

NASA

ULTRASONIC VELOCITY MEASUREMENT USING PHASE-SLOPE AND CROSS-CORRELATION METHODS

David R. Hull, Harold E. Kautz, and Alex Vary
National Aeronautics and Space Administration
Lewis Research Center
Cleveland, Ohio 44135

SUMMARY

Computer implemented phase-slope and cross-correlation methods are introduced for measuring time delays between pairs of broadband ultrasonic pulse-echo signals for determining velocity in engineering materials. The phase-slope and cross-correlation methods are compared with the overlap method which is currently in wide use. Comparison of digital versions of the three methods shows similar results for most materials having low ultrasonic attenuation. However, the cross-correlation method is preferred for highly attenuating materials. An analytical basis for the cross-correlation method is presented. Examples are given for the three methods investigated to measure velocity in representative materials in the megahertz range.

INTRODUCTION

Ultrasonic velocity measurements are widely used to determine properties and states of materials. In the case of engineering solids measurements of ultrasonic wave propagation velocities are routinely used to determine elastic constants (refs. 1 to 5). There has been an increasing use of ultrasonic velocity measurements for nondestructive characterization of material microstructures and mechanical properties (refs. 6 to 10). Therefore, it is important to have appropriate practical methods for making velocity measurements on a variety of material samples.

A number of approaches to the measurement of ultrasonic velocity in solids have been described (refs. 11 and 12). Included among the approaches there are those that use single frequency continuous waves, tone bursts and broadband pulse-echo waveforms. In the latter case, the basic problem is to determine the time delay between two successive echoes, each of which consists of only a few oscillations. The pulse-echo approach using a single transducer in contact with the material sample has many practical uses (refs. 13 and 14). Therefore, it will be treated exclusively in this report.

In many instances it is fairly easy to find some feature in the pulse-echo waveforms (e.g., the principal oscillation) that can be used as a basis for measuring the time between echoes. The "pulse-echo overlap" method (ref. 11) is commonly used to make this type of measurement. The method can be implemented by analog or digital instrumentation (ref. 12). In either case, the echoes must have similar waveforms so that corresponding features can be readily identified and brought into coincidence (i.e., overlapped) either by analog or digital manipulations. Since this condition is not always met, alternative methods are needed.

The purpose of this report is to deal with those cases where pulse-echo signals are weak or distorted by attenuation and other factors that render them unsuitable for the overlap approach. This report describes ultrasonic velocity measurement methods based on computer digitization of broadband pulse-echo waveforms. Three digital methods are compared: (1) overlap, (2) phase-slope, and (3) cross-correlation. The analytical basis for the cross-correlation method is emphasized. Specific examples are given for observed signals in a frequency range from 5 to 50 MHz for metal and composite samples.

EQUIPMENT

Transducer

A typical transducer configuration for normal incidence, contact, pulse-echo ultrasonics appears in figure 1(a). In the transducer a piezoelectric crystal is excited by a high voltage drive pulse to produce a broadband longitudinal ultrasonic wave. This wave propagates through the quartz buffer to the front surface where a portion of the wave's energy is transmitted into the specimen and a portion is reflected back to the piezoelectric crystal. The portion of the ultrasonic wave reflected from the front surface is the first echo, FS (fig. 1(b)). The wave transmitted into the specimen is reflected by the back surface to the front surface where a portion is reflected back into the specimen and a portion is transmitted through the quartz buffer to the piezoelectric crystal. This wave is shown as echo B1 in figure 1(b). Finally a third ultrasonic wave, echo B2, occurring from a second back surface reflection is received at the piezoelctric crystal as shown in figure 1(b).

Instrumentation

Block diagrams of instrumentation used for this report appears in figures 2(a) and (b). The major components are: (1) ultrasonic pulser-receiver, (2) digitizing oscilloscope, and (3) minicomputer. Auxiliary components include a digital CRT monitor, real-time video monitor, digital time delay, and hard copier. The minicomputer operates under TEK SPS BASIC software (ref. 15) and various system components are interfaced with the IEEE 488 GPIB (General Purpose Interface Bus). The ultrasonic pulser-receiver is an analog device used to excite the transducer and to receive resultant echoes. Thereafter, signal acquisition, processing, and analysis are done digitally. Methods for digital processing described herein depend on use of the discrete Fourier transform algorithm. It is implemented by an FFT (fast Fourier transform) software package which is part of the system of figure 2.

PROCEDURE

General

In figure 1(b) time, T, between echoes B1 and B2 is that required for ultrasonic pulses to travel twice the thickness, s, of the sample. The time between echoes FS and B1 is also T. Here echo FS will be ignored because it is preferable to use echoes from within the specimen. The procedure for calculating velocity, $v = 2s/T$, is based on analysis of the two back surface echoes B1 and B2. All examples given are based on longitudinal wave propagation.

The initial step in the procedure for signal analysis requires digitization of echoes B1 and B2. Two typical cases can arise: In the first case, both echoes are displayed simultaneously on one CRT trace (as in fig. 2(a)). In the second case, the echoes are individually "windowed" in two separate traces (as in fig. 2(b)). These "windows" are time gates used to isolate waveforms of interest. In digitized form they usually consist of 512-element arrays with 9 bit vertical resolution. The individually "windowed" case is used when there is a long time interval between echoes. Then, any attempt to compress two echoes into a single CRT trace can significantly reduce the amount of usable detail in the individual echoes. The signals could also be rendered useless due to aliasing, according to the Nyquist criterion for digitization (ref. 16).

When both echoes B1 and B2 cannot be simultaneously windowed, a digital delay device is used. It introduces a known time delay between the beginnings of the two separate windows containing the waveforms for echoes B1 and B2. The time delay device is calibrated to an accuracy of ±1 ns and has a precision of ±0.1 ns. As indicated in figure 2(b), this delay is triggered by synchronization signals from the pulser-receiver. After B1 is windowed and digitized, a predetermined delay is generated to put echo B2 into the digitizing window.

The process is shown in figure 3 for a case where the time, T, between echoes B1 and B2 is calculated from $T = W + (T2 - T1)$, where, W is the time delay between windows and times T1 and T2 are in-window delays measured with respect to corresponding major oscillations of echoes B1 and B2, respectively. If both echoes can be included in the same window, $W = 0$ and the expression reduces to $T = (T2 - T1)$.

Overlap Method

Figure 4 shows the result of overlapping the echoes B1 and B2 from figure 3. The process involves a digital search for the peak voltage and the array element that contains it for each echo's major oscillation. Since, except for amplitude, the echoes have similar waveforms the peak value criterion is adequate to overlap the echoes.

The overlap method depends on having a pair of echoes, B1 and B2, that exhibit similar waveforms with corresponding features. But, even for the case illustrated in figure 4, it is apparent that the overlap is imperfect, e.g., the zero crossings are not all coincident. Effects of wave distortions due to noise, dispersion, and other factors that operate on successive echoes are present. The ability to deal with these distortion effects is facilitated by operating in the frequency rather than the time domain. It will be seen that frequency domain methods (phase-slope and cross-correlation) have advantages over the time domain method.

Phase-Slope Method

With the phase-slope method time between echoes is found by use of phase spectra of echo waveforms. After the echoes are digitized, a Fourier transform of each is obtained by a discrete FFT algorithm. The amplitude and continuous phase spectra for a pair of typical echoes are illustrated in figure 5.

After Fourier transformation, both the amplitude and phase spectra are used to define a central zone within the frequency domain. For example, this zone may consist of only a narrow range near the center frequency or a frequency range for which the amplitude exceeds some fraction of the peak value. And/or, the zone may consist only of the frequency range for which the phase spectrum is linear. These restrictions eliminate the low and high frequency extremes where the signal-to-noise ratio is low.

The group velocity is given by $U(f) = 2\pi(2s)/((d\theta/df) - W)$ where f is frequency, θ is phase angle in radians and W is the time delay between echo windows (ref. 17). In window time delays for each echo can be calculated by setting $W = 0$ and solving for $T(f) = (d\theta/df)/(2\pi)$. In the present case the phase spectra are linear functions of frequency in the central zone, their slopes are constants; $M = d\theta/df$, and the in-window time delays are $T1 = M1/(2\pi)$ for echo B1 and $T2 = M2/(2\pi)$ for echo B2. The total time delay $T = W + (T2 - T1)$.

Cross-Correlation Method

The digital cross-correlation method eliminates the need for the somewhat arbitrary criteria, e.g., peak value and zone for phase slope, applied in the two previously described methods. Unlike the overlap or phase-slope methods, cross-correlation does not require explicit criteria for accepting or rejecting specific features in echoes affected by distortion or low signal-to-noise ratios.

The method is illustrated in figure 6 which shows the normalized cross-correlation function for three cases of echo positions. The cross-correlation function of two time domain signals is obtained by conjugate multiplication using their Fourier transforms in the frequency domain and retransformation back to the time domain (ref. 16). The cross-correlation function possesses a maximum in the time domain. The displacement of this maximum relative to a zero reference gives the time interval C, which for the ideal case should equal $T2 - T1$ as measured by the digital overlap method. Cross-correlation gives this quantity whether the echoes appear in the same or separate windows. If the echoes B1 and B2 are separated windowed, then $T = W + C$.

For the relatively simple and undistorted echoes of figure 6 cross-correlation gives results similar to those obtained by the two previous methods. The validity of the cross-correlation method depends on the fact that the displacement of the maximum of the cross-correlation function equals the time delay between the two successive echoes. This is supported by the previous examples and the theory and results presented hereinafter.

THEORY

The theoretical analysis proceeds according to the following logic. First, it is noted that the center value of an auto-correlation function is always a maximum. Then, it is shown that the cross-correlation function can be rewritten as an auto-correlation function, this rewritten form has a maximum displaced from its center and that the displacement corresponds to the time delay sought

The cross-correlation function $R_{xy}(\tau)$ is defined in reference 18 as,

$$R_{xy}(\tau) = \lim_{T\to\infty} (1/T) \int_0^T X(t)Y(t + \tau)dt \qquad (1)$$

where, $X(t)$ and $Y(t)$ are arbitrary waveforms, where t is time and τ is a delay interval. If $Y(t)$ is replaced by $X(t)$, we have the auto-correlation function,

$$R_{xx}(\tau) = \lim_{T\to\infty} (1/T) \int_0^T X(t)X(t + \tau)dt \qquad (2)$$

It is assumed that the data of interest are stationary, i.e., that the time averages of equations (1) and (2) exist. It is shown in reference 18 that $R_{xx}(0) \geq R(\tau)_{xx}$ for all τ. That is, the center value of the auto-correlation function at $\tau = 0$ is always a maximum.

In an analysis of propagation paths (refs. 16 and 18) (as those associated with echoes B1 and B2) a function H_k can be defined. This function operates along the k^{th} propagation path on the input signal $X(t)$. The resultant output signal $Y(t)$ is,

$$Y(t) = \sum_k H_k X(t) + A(t) \qquad (3)$$

where $A(t)$ represents accumulated random noise. Equation (3) applies to pulse echoes (as in fig. 1(b)) if the input function $X(t)$ is taken as the first echo B1. Subsequent echoes such as B2 are functions of the propagation function H_k. By excluding higher order echoes, equation (3) reduces to,

$$Y(t) = HX(t) + A(t) \qquad (4)$$

In the case illustrated in figure 1(b) the function H operates on the input signal $X(t)$, or echo B1, in the following ways before it yields $Y(t)$, or echo B2. First, it attenuates echo B2 relative to echo B1. Second, it may invert echo B2 relative to echo B1 (e.g., reflection at a free back surface, i.e., multiplication by -1) followed by a second inversion at the buffer-sample interface, reinverting B2 to give results as in figure 1(b)). Third, it delays echo B2 relative to echo B1 by the round trip delay time in the sample. The delay time, T, is related to the propagation velocity, v, and sample thickness, s, by $T = 2s/v$. The effect of H operating on $X(t)$ can be expressed as:

$$HX(t) = \pm DB(f)X(t) \qquad (5)$$

The $\pm$ sign on the right side of equation (5) corresponds to inversions of echo B2 relative to echo B1 and depends on reflection coefficients at the buffer/sample interface. The time delay D is an operator changing t to $t - 2s/v$. B is the attenuation operator which is, in general a function of frequency f. B must operate in the frequency domain as shown:

$$BX(t) = F^{-1}\,[BF(X(t))] \tag{5a}$$

Here F is the Fourier transform and F^{-1} is the inverse Fourier transform, defined so that:

$$F^{-1}\,[F(X(t))] = X(t) \tag{5b}$$

If, however, B does not distort the shape of $X(t)$ too much, then we can approximate it by a constant b and write;

$$\begin{aligned} Y(t) = HX(t) &= \pm bDX(t) \\ &= \pm bX(t - 2s/v) \end{aligned} \tag{6}$$

According to reference 15, the discrete cross-correlation function for a digitized finite waveform can be written as,

$$R_{xy}(n) = (1/N) \sum_{t=0}^{N-n-1} X(t)Y(n + t) \tag{7}$$

where N equals the number of data sampling intervals corresponding to the delay time T. The quantity n ranges from $-(N - 1)$ to $+(N - 1)$. Combining equations (6) and (7) for the cross-correlation of echoes B1 and B2 gives,

$$R_{xy}(n) = (\pm)b/N \sum X(t)X[n + t - (2s/v)] \tag{8}$$

In reference 16 it is shown that equation (8) is equivalent to the autocorrelation of $X(t)$ with the correlation array points shifted to the right a distance $2s/v$. Thus the central maximum, or minimum, is shifted to the point a distance $2s/v$ right of center, in the case of a time lag (delay) from the first to second echo (as in fig. 6(b)).

EXAMPLE

The previous examples of digital methodology (figs. 3 to 6) were based on a metal sample (250-grade maraging steel). The velocities, measured with a 50 MHz transducer, are compared in table I for each of the three digital methods used. As expected, the results agree quite closely for the relatively clean, uncomplicated waveforms involved.

A further experimental example involving noisy, complicated waveforms is illustrated in figure 7. In this case both echoes, B1 and B2, occur in the same window. The signals were obtained from a graphite/epoxy composite laminate using a 5 MHz transducer. Due to the nature of the material, there is considerable attenuation, distortion, and noise. In addition, echo B2 is inverted relative to echo B1 because of the sign of the reflection coefficient at the buffer/sample interface.

The example in figure 7 demonstrates the facility of the cross-correlation method. Figure 7(a) shows echoes B1 and B2 and figure 7(b) shows the corresponding cross-correlation function. In figure 7(b) the maximum is displaced from 0 by 0.750 ± 0.002 μs. It is negative due to the inversion of echo B2.

Figure 7(c) exhibits highly distorted and noisy echoes obtained by windowing the third and fourth back surface echoes, B3 and B4, without removing the specimen from the transducer. The time between this pair should be identical to that between echoes B1 and B2. Therefore, the result of applying the cross-correlation method in figure 7(d) can be compared directly with the previously determined value for time T. For figure 7(d) $T = 0.752 \pm 0.002$ μs with the difference between the two within the timing error corresponding to the data sampling interval. However, the digital overlap method gave $T = 0.748 \pm 0.002$ μs for echoes B1 and B2 and $T = 0.766 \pm 0.002$ μs for echoes B3 and B4. The computer program used to the results in figure 7 is listed in the appendix.

Even in the case of clean, uncomplicated echoes, all three methods are subject to a "data sampling" error peculiar to the digitization process. This is due to the finite number of discrete data points that the digitization process assigns to the waveforms collected. A random error tends to arise from a variety of sources, e.g., small jittering effects in the oscilloscope sweeps. To estimate this inherent, random error, a series of 30 waveform pairs were digitized, in sequence, at the same location on a material sample. The results appear in table I which compares the three digital methods for a metal sample and a composite sample. Typical errors in the time delay are given in table I for the particular material samples, frequencies, and signal processing methods used in this study. Note that the time delay errors are compared for a fixed thickness of each material. It is seen in table I that the cross-correlation method gives the lowest random time delay errors for both samples. The phase-slope method on the composite sample produced the highest random time delay error due to the low signal-to-noise ratio of the echoes.

DISCUSSION

The digital overlap method is an effective way to automate the analog overlap method. The overlap method requires input of information concerning echo inversions, peak selection, etc. Once overlap criteria are established, the digital time domain method simulates a manual overlap based on visual inspection of analog waveforms. However, even digital signal manipulation of time domain signals can be problematic if the waveforms are subject to noise, attenuation, and similar factors.

The frequency domain phase-slope method eliminates problems encountered in the time domain, e.g., the need to account for echo inversions. In addition, it provides convenient criteria for selecting an appropriate frequency range in cases where the major portion of the phase spectra of the echoes are mutually linear. Generally, for nonlinear dispersive cases, the phase-slope method can determine group velocities as functions of frequency (ref. 17). However, if signal-to-noise ratio is low, as was the case for the composite sample in table I, poor results are obtained.

The advantages of the cross-correlation method are apparent when the signal-to-noise ratio is low and/or random noise is superimposed on the echoes. One of the demonstrated properties of the cross-correlation function is that it is (statistically) weighted by dominant frequencies common to the waveforms being correlated (refs. 16 and 18). Therefore, the cross-correlation function returns a group velocity within the frequency bandwidth of the signals analyzed. With the cross-correlation method, effects of random noise, extraneous

signals, etc. are minimized. The example of figures 7(c) and (d) illustrates the utility of the cross-correlation method for highly attenuating and noisy materials.

CONCLUSION

Three computer implemented digital methods for measuring ultrasonic velocity in solids are described: the overlap, phase-slope, and cross-correlation methods. It is evident from the examples given that the cross-correlation method is the most versatile of the three digital methods for measuring time delays between broadband ultrasonic pulse-echo signals. All three methods produce comparable results when applied to pairs of clean, undistorted echoes. The cross-correlation method yields the most reliable and reproducible results in the case of highly attenuating materials that return very noisy, distorted echoes. A discussion of theory and relevant properties of the cross-correlation function indicates the basis for the versatility and utility of the cross-correlation method.

APPENDIX

PROGRAM LISTING

The program listing given here is written in TEK SPS BASIC for implementation with a Tektronix 7854 digital oscilloscope and 4052 minicomputer (ref. 7). The instrument configuration is that shown in figure 2(a) using a Panametrics 5052PR pulser-receiver and 5 MHz transducer. This information is given only for illustrative purposes and does not represent an endorsement of the commercial equipment or software named.

```
100 REM  CALCULATION OF VELOCITY BY USE OF CROSS-CORRELATION
105 REM
110 REM  INSTRUMENTATION:  TEKTRONIX 7854 DIGITAL OSCILLOSCOPE
115 REM                          4052 COMPUTER
120 REM                    ROM PACKS R07 SIGNAL PROCESSING No.1
125 REM                              R08 SIGNAL PROCESSING No.2 (FFT)
130 REM                    PANAMETRICS 5052PR PULSER RECEIVER
135 REM                    QUARTZ BUFFERED TRANSDUCER
140 REM
145 REM   ------------------------------------------------------------
150 REM   !                      INITIAL INPUTS                        !
155 REM   ------------------------------------------------------------
160 PRINT "ENTER DATE [DD-MMM-YY] : ";
165 INPUT D$
170 PRINT "ENTER TRANSDUCER FREQUENCY : ";
175 INPUT T$
180 PRINT "ENTER SPECIMEN ID. : ";
185 INPUT S$
190 PRINT "ENTER THICKNESS IN cm = ";
195 INPUT T
200 REM   ------------------------------------------------------------
205 REM   !             COLLECTION OF WAVEFORM FROM 7854               !
210 REM   ------------------------------------------------------------
215 ON SRQ THEN 905
220 PRINT "POSITION ULTRASONIC WAVEFORM ON 7854 WITH"
225 PRINT "B1 left OF CENTER AND B2 right OF CENTER."
230 PRINT "PRESS <RETURN> TO COLLECT WAVEFORM !!"
235 INPUT Z$
240 V$="EXEC OFF 1 0 2 4 >P/W STORED 1 0 AVG 0 WFM MEAN - 0 WFM SENDX"
245 PRINT @1:V$
250 INPUT @1:A1,A2,A3,A4,A5
255 DELETE A
260 DIM A(A1)
265 INPUT @1:A
270 A=A*A5
275 A=A+A4
280 V$="V"
285 H$="uS"
290 PRINT @1:"SCOPE"
295 REM   ------------------------------------------------------------
300 REM   !              GO PLOT ULTRASONIC WAVEFORM                   !
305 REM   ------------------------------------------------------------
310 PAGE
315 VIEWPORT 60,120,60,95
320 GOSUB 930
325 REM   ------------------------------------------------------------
330 REM   !   BREAK ARRAY IN HALF TO PREPARE FOR CROSS-CORRELATION     !
335 REM   ------------------------------------------------------------
340 A1=A1/2
345 DELETE X,Y
350 DIM X(A1),Y(A1)
355 FOR J=1 TO A1
360 X(J)=A(J)
365 Y(J)=A(J+A1)
370 NEXT J
375 REM   ------------------------------------------------------------
380 REM   !              FIND MAX AND MIN OF TIME DOMAIN               !
385 REM   ------------------------------------------------------------
390 CALL "MAX",X,M1,C1
395 CALL "MAX",Y,M2,C2
400 CALL "MIN",Y,M3,C3
405 C2=C2+A1
410 C3=C3+A1
```

```
415 REM   ------------------------------------------------------------
420 REM   | PERFORM FOURIER TRANSFORM ON EACH HALF FOR PHASE SLOPE |
425 REM   ------------------------------------------------------------
430 DELETE B1,B2,P1,P2,M
435 DIM B1(A1),B2(A1),P1(A1/2+1),P2(A1/2+1),M(A1/2+1)
440 B1=X
445 B2=Y
450 CALL "FFT",B1
455 CALL "FFT",B2
460 CALL "POLAR",B1,M,P1,0
465 CALL "POLAR",B2,M,P2,0
470 CALL "DIF3",P1,P1
475 CALL "DIF3",P2,P2
480 CALL "MAX",M,P4,P3
485 P4=1/(A3*A1)
490 IF P1(P3)<0 THEN 500
495 P1=P1-2*PI
500 IF P2(P3)<0 THEN 525
505 P2=P2-2*PI
510 REM   ------------------------------------------------------------
515 REM   |        PERFORM CROSS-CORRELATION, FIND MAX & MIN         |
520 REM   ------------------------------------------------------------
525 CALL "CORR",X,Y,A
530 DELETE X,Y
535 CALL "MAX",A,M8,C8
540 CALL "MIN",A,M9,C9
545 IF -M9<M8 THEN 555
550 M8=-M9
555 A=A/M8
560 V$=""
565 CALL "MAX",A,M4,C4
570 CALL "MIN",A,M5,C5
575 I$=""
580 REM   ------------------------------------------------------------
585 REM   |  MAKE DECISION ON INVERSION & SET APPROPRIATE VARIABLES |
590 REM   ------------------------------------------------------------
595 IF M4>ABS(M5) THEN 620
600 I$="(INVERSION)"
605 M4=M5
610 C4=C5
615 C2=C3
620 MOVE C1,M1
625 DRAW C1,M1*1.15
630 MOVE C2,M2
635 DRAW C2,M2*1.15
640 REM   ------------------------------------------------------------
645 REM   |  GO PLOT CROSS-CORRELATION WAVEFORM & DRAW DELAY MARK  |
650 REM   ------------------------------------------------------------
655 A1=A1*2
660 VIEWPORT 60,120,10,40
665 GOSUB 930
670 A1=A1/2
675 MOVE C4,M4*1.1
680 DRAW C4,M4*1.5
685 PRINT USING "1A,1D.4D,3A":"(",C4*A3*1000000," uS"
690 T1=(C2-C1)*A3*1000000
695 T2=(A1*A3+(P1(P3)-P2(P3))/P4/(2*PI))*1000000
700 T3=C4*A3*1000000
705 V1=2*T/T1
710 V2=2*T/T2
715 V3=2*T/T3
720 REM   ------------------------------------------------------------
725 REM   |         RESET VIEWPORT & WINDOW, PRINT OUT RESULTS      |
730 REM   ------------------------------------------------------------
735 VIEWPORT 0,130,0,100
740 WINDOW 0,130,0,100
745 MOVE 73,98
750 PRINT "ULTRASONIC WAVEFORM"
755 MOVE 70,43
760 PRINT "CROSS-CORRELATION ";I$
765 MOVE 0,98
770 PRINT "DATE: ";D$
775 PRINT "JJSPECIMEN: ";S$
780 PRINT "THICKNESS= ";T;" cm"
785 PRINT "JTRANSDUCER: ";T$
790 PRINT "JMETHOD - VELOCITY, DELAY,"
```

```
795 PRINT "          cM/uS      uSJ"
800 PRINT USING "7A,2A,1D.5D,2A,1D.5D":"OVERLAP","  ",V1,"  ",T1
805 PRINT USING "7A,2A,1D.5D,2A,1D.5D":"PHASE  ","  ",V2,"  ",T2
810 PRINT USING "7A,2A,1D.5D,2A,1D.5D":"CORR   ","  ",V3,"  ",T3
815 PRINT "JJ";A1;" POINTS / WAVEFORM"
820 PRINT USING "16A,1D.4D,3A":"TIMING ERROR +U_",A3*1000000," uS"
825 PRINT USING "6A,1D.2D,2A":"OR +H_",1/C4*100," %"
830 REM   ------------------------------------------------------------
835 REM   !    SHOW OPTIONS & MAKE DECISION ABOUT PROGRAM FLOW    !
840 REM   ------------------------------------------------------------
845 PRINT "JJJJJJJINPUT CHOICE SHOWN BELOW:J"
850 PRINT " 'R' TO REPEAT SAME SAMPLE"
855 PRINT "'N' TO START A NEW SAMPLE"
860 PRINT "'S' OR (ENTER) TO STOP!"
865 PRINT "JCHOICE : GG";
870 INPUT Z$
875 IF Z$="R" THEN 215
880 IF Z$="N" THEN 180
885 END
890 REM   ------------------------------------------------------------
895 REM   !          SUBROUTINE #1 -- ON SRQ POLL CHECK            !
900 REM   ------------------------------------------------------------
905 POLL D,S;1
910 RETURN
915 REM   ------------------------------------------------------------
920 REM   !        SUBROUTINE #2 -- DRAW AXES & WAVEFORM           !
925 REM   ------------------------------------------------------------
930 CALL "MIN",A,M1,L1
935 CALL "MAX",A,M2,L2
940 W1=INT(LGT(A3*A1))
945 W2=A1
950 IF ABS(M1)<M2 THEN 965
955 M2=-M1
960 GO TO 970
965 M1=-M2
970 W3=LGT(M2-M1)
975 W3=INT(W3)
980 W4=INT((M2-M1)/10↑W3)+1
985 M3=-W4/2
990 WINDOW 0,W2,M3*10↑W3,(W4+M3)*10↑W3
995 AXIS W2/10,W4*10↑W3/8,0,M3*10↑W3
1000 Z$=STR(10↑W3)
1005 MOVE 1,(W4+M3)*10↑W3
1010 PRINT "KHHHHHHHHHHHHH";Z$;" ";V$
1015 FOR I=0 TO 8 STEP 2
1020 MOVE 1,W4*10↑W3/8*I+M3*10↑W3
1025 PRINT USING " ""HHHHHHHH"",4D.2D":W4/8*I+M3
1030 NEXT I
1035 Z$=STR(10↑W1*1000000)
1040 MOVE A1/1.35,M3*10↑W3*1.1
1045 PRINT "JJ";Z$;" ";H$;
1050 FOR I=0 TO 10 STEP 2
1055 MOVE I*W2/10,M3*10↑W3*1.05
1060 PRINT USING """HHJ"",2D.1D,S":I*W2/10*A3/10↑W1
1065 NEXT I
1070 CALL "DISP",A
1075 RETURN
```

REFERENCES

1. Schreiber, E., O. L. Anderson, and N. Soga, Elastic Constants and Their Measurement, 1973, McGraw Hill, New York, NY.

2. Truell, R., C. Elbaum, and B. B. Chick, Ultrasonic Methods in Solid State Physics, 1969, Academic Press, New York, NY.

3. Kim, B. S., "Material Property Measurements of PZT-5A-Type Piezoelectric Ceramics," Materials Evaluation, Vol. 40, Oct. 1982, pp. 1180-1183.

4. Lynnworth, L. C., "Ultrasonic Measurements of Elastic Moduli in Slender Specimens Using Extensional and Torsional Wave Pulses," Journal of Testing and Evaluation, Vol. 1, Mar. 1973, pp. 119-125.

5. Speich, G. R., A. J. Schwoeble, and W. C. Leslie, "Elastic Constants of Binary Iron-Base Alloys," Metallurgical Transactions, Vol. 3, Aug. 1972, pp. 2031-2037.

6. Klinman, R. and E. T. Stephenson, "Ultrasonic Prediction of Grain Size and Mechanical Properties in Plain Carbon Steel," Materials Evaluation, Vol. 39, Nov. 1981, pp. 1116-1120.

7. Grayeli, N., D. B. Ilic, F. Stanke, G. S. Kino, and J. C. Shyne, "Acoustic Measurement of Microstructure in Steels," DARPA/AMFL Review of Progress in Quantitative NDE, AFWAL-TR-80-4078, July, 1980, pp. 429-436.

8. Papadakis, E. P., "Ultrasonic Attenuation and Velocity in SAE 52100 Steel Quenched from Various Temperatures," Metallurgical Transactions, Vol. 1, Apr. 1970, pp. 1053-1057.

9. Vary, A., "Ultrasonic Measurement of Material Properties," Research Techniques in Nondestructive Testing, Vol. 4, 1980, pp. 159-204, Academic Press, London, England.

10. Green, R. E., Jr., Ultrasonic Investigation of Mechanical Properties, 1973, Academic Press, New York, NY.

11. Papadakis, E. P., "Ultrasonic Velocity and Attenuation Measurement Methods with Scientific and Industrial Applications," Physical Acoustics: Principles and Methods, Vol. 12, 1976, pp. 277-374, Academic Press, New York, NY.

12. Vary, A., "Computer Signal Processing for Ultrasonic Attenuation and Velocity Measurements for Material Property Characterization," Proceedings of the Twelfth Symposium on Nondestructive Evaluation, 1979, pp. 33-46. Southwest Research Institute, San Antonio, TX. (NASA TM-79180).

13. Papadakis, E. P., "Ultrasonic Attenuation and Velocity in Three Transformation Products in Steel," Journal of Applied Physics, Vol. 35, May 1964, pp. 1474-1482.

14. Arons, R. M. and D. S. Kupperman, "Use of Sound-Velocity Measurements to Evaluate the Effect of hot Isostatic Pressing on the Porosity of Ceramic Solids," Materials Evaluation, Vol. 40, Sept. 1982, pp. 1076-1078.

15. TEKTRONIX 4050 Series, R08, Signal Processing ROM Pack 2 (FFT), Manual, Part No. 070-2841-00, Dec. 1980. Tektronix, Inc., Beaverton, OR.

16. Bendat, J. S. and A. G. Piersol, Engineering Applications of Correlation and Spectral Analysis, 1980, Wiley, New York, NY.

17. Sachse, W. and Y. H. Pao, "On the Determination of Phase and Group Velocities of Dispersive Waves in Solids," Journal of Applied Physics, Vol. 49, Aug. 1978, pp. 4320-4327.

18. Bracewell, R. N., The Fourier Transform and Its Applications, 1978, McGraw-Hill, New York, NY.

TABLE I. - COMPARISON OF DIGITAL METHODS FOR ULTRASONIC VELOCITY MEASUREMENT

Method	Metal sample[a]			Composite sample[b]		
	Time delay, T, μs	Random[c] time delay error, ±μs	Ultrasonic velocity, v, cm/μs	Time delay, T, μs	Random[c] time delay error, ±μs	Ultrasonic velocity, v, cm/μs
Overlap	0.90873	0.00012	0.5500	0.74850	0.00322	0.3006
Phase-slope	.90922	.00005	.5497	.72417	.01963	.3107
Cross-correlation	.90873	[d]0	.5500	.74975	.00049	.3001

[a]Metal sample: 250-grade maraging steel. Thickness: 2.499 mm. Estimated error in velocity: ±0.2 percent. Measured with a 50 MHz center frequency transducer.
[b]Composite sample: graphite/epoxy, 8-ply, (0 ± 45) symmetrical. Thickness: 1.125 mm. Estimated error in velocity: ±0.2 percent. Measured with a 5 MHz center frequency transducer.
[c]Time delay errors are based on standard deviation of time delay for 30 echo pairs at fixed location on material sample.
[d]The cross-correlation method calculated the time delays to be identical for all 30 pairs of echoes.



(a) Cross-section showing principle echoes.



(b) Time domain trace of principle echoes.

Figure 1. - Specimen-transducer configuration for pulse-echo velocity measurements and principal echoes. Echo FS is returned by front surface of specimen and B1 and B2 are successive echoes returned by the back surface. Time, T, is the "round trip" time delay between echoes B1 and B2.



(a) Schematic diagram for acquiring two successive echoes, B1 and B2, in the same time window.

Figure 2. - Diagrams for two alternative waveform acquisition schemes.



(b) Schematic diagram for acquiring two successive echoes, B1 and B2, in separate time windows.

Figure 2. - Concluded.



Figure 3. - Representative back echoes B1 and B2 separately windowed shown in digitized form. Time delay, W, between windows is a predetermined, known quantity. Time delay T, between echoes B1 and B2 is calculated from T = W + (T2 - T1) by using one of the methods illustrated in figures 4, 5, and 6.



Figure 4. - Result of digital overlap method for determining delay between echoes B1 (solid) and B2 (dotted) using echoes shown in figure 3.



Figure 5. - Typical back surface echoes, B1 (solid) and B2 (dotted), and their amplitude and phase spectra. In-window time delays based on the digital overlap method gives T1 = 39.8 nsec and T2 = 91.0 nsec. The phase-slope method gives T1 = 38.9 nsec and T2 = 90.7 nsec. Differences between the time delays can be attributed to the data sampling interval error of 0.4 nsec.



Figure 6. - Result of generating the normalized cross-correlation function for three cases of echo positions. The displacement of the absolute maximum of the cross-correlation function relative to the zero reference gives the time delay C. For these ideal echoes C was equal to T2 - T1 as measured by digital overlap within the data sampling interval error of 0.4 nsec.



(b) T1 < T2.

Figure 6. - Continued.



(c) T1 > T2.

Figure 6. - Concluded.



(a) Time domain trace of echoes B1 and B2. T2 - T1 = 0.748 ± 0.002 μsec.

(b) Normalized cross-correlation function for B1 versus B2. C = 0.750 ± 0.002 μsec.

(c) Time domain trace of echoes B3 and B4 T2 - T1 = 0.766 ± 0.002 μsec.

(d) Normalized cross-correlation function for B3 versus B4. C = 0.752 ± 0.002 μsec.

Figure 7. - Examples of digital cross-correlation method for determining time delay between simultaneously windowed, successive echo pairs exhibiting increasingly low signal-to-noise ratios. Backsurface echoes for a graphite epoxy specimen using a 5 MHz transducer. Each pair of echoes is the average of 10 waveforms.

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM-83794		

4. Title and Subtitle	5. Report Date
Ultrasonic Velocity Measurement Using Phase-Slope and Cross-Correlation Methods	6. Performing Organization Code: 506-53-1A

7. Author(s)	8. Performing Organization Report No.
David R. Hull, Harold E. Kautz, and Alex Vary	E-2290
	10. Work Unit No.

9. Performing Organization Name and Address	11. Contract or Grant No.
National Aeronautics and Space Administration Lewis Research Center Cleveland, Ohio 44135	13. Type of Report and Period Covered: Technical Memorandum

12. Sponsoring Agency Name and Address	14. Sponsoring Agency Code
National Aeronautics and Space Administration Washington, D.C. 20546	

15. Supplementary Notes

Prepared for the 1984 Spring Conference of the American Society for Nondestructive Testing, Denver, Colorado, May 21-24, 1984.

16. Abstract

Computer implemented phase-slope and cross-correlation methods are introduced for measuring time delays between pairs of broadband ultrasonic pulse-echo signals for determining velocity in engineering materials. The phase-slope and cross-correlation methods are compared with the overlap method which is currently in wide use. Comparison of digital versions of the three methods shows similar results for most materials having low ultrasonic attenuation. However, the cross-correlation method is preferred for highly attenuating materials. An analytical basis for the cross-correlation method is presented. Examples are given for the three methods investigated to measure velocity in representative materials in the megahertz range.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Ultrasonic velocity; Cross-correlation function; Phase-slope; Group velocity; Nondestructive testing	Unclassified - unlimited STAR Category 38

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of pages	22. Price*
Unclassified	Unclassified		

*For sale by the National Technical Information Service, Springfield, Virginia 22161

National Aeronautics and
Space Administration

Washington, D.C.
20546

Official Business
Penalty for Private Use, $300

SPECIAL FOURTH CLASS MAIL
BOOK



Postage and Fees Paid
National Aeronautics and
Space Administration
NASA-451

NASA

POSTMASTER: If Undeliverable (Section 158
Postal Manual) Do Not Return